Second SPICE Exercise

Fundamentals Of Electronics - a.a. 2018-2019 - University of Padua (Italy)

Pietro Prandini (mat. 1097752)

June 4, 2019

Contents

1 NMOS common source amplifier with bypass capacitance **5**
1.1 Analytic solutions . . . 5
1.1.1 DC analysis . . . 5
R_D . . . 5
R_S . . . 6
V_{GS} . . . 6
R_{G1} . . . 6
g_m . . . 7
r_0 . . . 7
1.1.2 AC analysis . . . 7
Hybrid π model . . . 7
R_{IN} from G . . . 7
R_{OUT} from D . . . 8
Voltage Gain - without R_{sig} and R_L . . . 8
Voltage Gain - with R_{sig} and R_L . . . 8
1.2 SPICE simulations . . . 12
1.2.1 DC simulation - Operating Point . . . 12
1.2.2 AC simulation - Av, R_{IN} and R_{OUT} . . . 12
1.2.3 AC simulation - Gv . . . 13

2 NMOS common source amplifier without bypass capacitance **15**
2.1 Analytic solution . . . 15
Hybrid π model . . . 15
R_{IN} from G . . . 15
R_{OUT} from D . . . 16
Voltage Gain - without R_{sig} and R_L . . . 16
Voltage Gain - with R_{sig} and R_L . . . 16
2.2 SPICE simulations . . . 20
2.2.1 AC simulation without bypass capacitances - Av, R_{IN} and R_{OUT} . . . 20
2.2.2 AC simulation without bypass capacitances - Gv . . . 20

Chapter 1

NMOS common source amplifier with bypass capacitance



Figure 1.1: NMOS common source amplifier

Designing the common source amplifier of the figure 1.1 .
The MOSFET should have a $V_t = 1V$, a $K_n = 4mA/V$ and a $\lambda = 0$.
Other requested parameters are: $I_{DQ} = 0.5mA$, $V_S = 3.5V$, $V_D = 11V$, $V_{DD} = 15V$ and $R_{G2} = 1097752\Omega$.

1.1 Analytic solutions

1.1.1 DC analysis

On a Direct Current analysis the capacitances can be considered as open circuits, the inductances can be considered as short circuits, the signal and the load are removed and the alternate current inputs are not considered.
The figure 1.2 represents the circuit for the DC analysis.

R_D

$$V_D = V_{DD} - R_D I_D \tag{1.1}$$

$$R_D = \frac{V_{DD} - V_D}{I_D} \tag{1.2}$$

$$R_D = \frac{15V - 11V}{0.5mA} = 8k\Omega \tag{1.3}$$



Figure 1.2: NMOS common source amplifier - DC analysis

R_S

$$V_S = R_S I_D \implies R_S = \frac{V_S}{I_D} \tag{1.4}$$

$$R_S = \frac{3.5V}{0.5mA} = 7k\Omega \tag{1.5}$$

V_{GS}

$$I_D = \frac{1}{2} K_n V_{ov}^2 \implies V_{ov} = \pm\sqrt{\frac{2I_D}{K_n}} \tag{1.6}$$

$$V_{ov} = \pm\sqrt{\frac{2 \cdot 0.5mA}{4mA/V^2}} \tag{1.7}$$

$$V_{ov} = \begin{cases} +0.5V & \text{Real value of } V_{ov}. \\ -0.5V & \text{No physical sense.} \end{cases} \tag{1.8}$$

$$V_{ov} = V_{GS} - V_t \implies V_{GS} = V_{ov} + V_t \tag{1.9}$$

$$V_{GS} = 0.5V + 1V = 1.5V \tag{1.10}$$

R_{G1}

$$V_{GS} = V_G - V_S \implies V_G = V_{GS} + V_S \tag{1.11}$$

$$V_G = 1.5V + 3.5V = 5V \tag{1.12}$$

$$I_G R_{G2} - V_{GS} - I_D R_S = 0 \implies I_G = \frac{V_{GS} + I_D R_S}{R_{G2}} \tag{1.13}$$

$$I_G = \frac{1.5V + 0.5mA \cdot 7k\Omega}{1097.752k\Omega} = 4.5547628\mu A \simeq 4.55\mu A \tag{1.14}$$

$$R_{G1} = \frac{V_{DD} - V_G}{I_G} \tag{1.15}$$

$$= \frac{15V - 5V}{4.5547628\mu A} \tag{1.16}$$

$$= 2.19550M\Omega \simeq 2.20M\Omega \tag{1.17}$$

g_m

$$g_m = K_n V_{ov} = 4mA/V^2 \cdot 0.5V = 2mA/V \tag{1.18}$$

r_0

$$r_0 = \frac{1}{\lambda I_D} \overset{\lambda=0}{\Longrightarrow} r_0 = \infty \quad r_0 \text{ is considered as an open circuit.} \tag{1.19}$$

1.1.2 AC analysis

On an Alternate Current analysis the capacitances can be considered as short circuits, the inductances can be considered as open circuits and the direct current inputs are not considered.
The figure 1.3 represents the circuit for the AC analysis.
Other requested parameters are: $R_{sig} = 200k\Omega$ and $R_L = 8k\Omega$.



Figure 1.3: NMOS common source amplifier - AC analysis

Hybrid π model

For a small signal analysis it can be used an equivalent model to represent the behaviour of the transistor.
In this case it is used the hybrid π model (figure 1.4).

R_{IN} from G

Removing the signal, the load and applying a test voltage source as in figure 1.5 it is possible to calculate the input's resistance R_{IN}.

$$R_{IN} = \frac{V_x}{I_x} \tag{1.20}$$

$$\tag{1.21}$$

$$I_x = \frac{V_x}{R_{G1} \parallel R_{G2}} \implies R_{G1} \parallel R_{G2} = \frac{V_x}{I_x} \implies R_{IN} = R_{G1} \parallel R_{G2} \tag{1.22}$$

$$R_{IN} = \frac{R_{G1} R_{G2}}{R_{G1} + R_{G2}} \tag{1.23}$$

$$= \frac{2.19550M\Omega \cdot 1097752\Omega}{2.19550M\Omega + 1097752\Omega} \tag{1.24}$$

$$= 733.16756k\Omega \simeq 733.2k\Omega \tag{1.25}$$



Figure 1.4: NMOS common source amplifier - Hybrid π model

R_{OUT} from D

Removing the signal, the load and applying a test voltage source as in figure 1.6 it is possible to calculate the output's resistance R_{OUT}.

$$R_{OUT} = \frac{V_x}{I_x} \tag{1.26}$$

$$\tag{1.27}$$

$$I_x = \frac{V_x}{R_D} \implies R_D = \frac{V_x}{I_x} \implies R_{OUT} = R_D \tag{1.28}$$

$$R_{OUT} = 8k\Omega \tag{1.29}$$

Voltage Gain - without R_{sig} and R_L

Calculating the gain of the amplifier represented in the figure 1.7.

$$v_{in} = v_{gs} \tag{1.30}$$

$$v_o = -g_m v_{gs} R_D \tag{1.31}$$

$$A_v = \frac{v_o}{v_{in}} = \frac{-g_m v_{gs} R_D}{v_{gs}} = -g_m R_D = 2mA/V \cdot 8k\Omega = -16 \quad V/V \tag{1.32}$$

Voltage Gain - with R_{sig} and R_L

Calculating the gain of the amplifier represented in the figure 1.8.

$$I_{sig} = \frac{v_{sig}}{R_{sig} + (R_{G1} \parallel R_{G2})} \tag{1.33}$$

$$v_{in} = v_{gs} = v_{sig} - R_{sig} I_{sig} \tag{1.34}$$

$$= v_{sig} - R_{sig} \frac{v_{sig}}{R_{sig} + (R_{G1} \parallel R_{G2})} \tag{1.35}$$

$$= v_{sig} \left(1 - R_{sig} \frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})} \right) \tag{1.36}$$



Figure 1.5: NMOS common source amplifier - Calculating R_{IN}

$$v_o = -g_m v_{gs}(R_D \parallel R_L) \tag{1.37}$$
$$= -g_m v_{sig}\left(1 - R_{sig}\frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right)(R_D \parallel R_L) \tag{1.38}$$

$$G_v = \frac{v_o}{v_{sig}} = \frac{-g_m v_{sig}\left(1 - R_{sig}\frac{1}{R_{sig}+(R_{G1}\parallel R_{G2})}\right)(R_D \parallel R_L)}{v_{sig}} \tag{1.39}$$
$$= -g_m\left(1 - R_{sig}\frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right)(R_D \parallel R_L) \tag{1.40}$$
$$= -g_m\left(1 - R_{sig}\frac{1}{R_{sig} + \left(\frac{R_{G1}R_{G2}}{R_{G1}+R_{G2}}\right)}\right)\left(\frac{R_D R_L}{R_D + R_L}\right) \tag{1.41}$$
$$= -2mA/V\left(1 - 200k\Omega\frac{1}{200k\Omega + \left(\frac{2.19550M\Omega\cdot 1097752\Omega}{2.19550M\Omega+1097752\Omega}\right)}\right)\left(\frac{8k\Omega\cdot 8k\Omega}{8k\Omega + 8k\Omega}\right) \tag{1.42}$$
$$= -6.28296 \quad V/V \simeq -6.3 \quad V/V \tag{1.43}$$



Figure 1.6: NMOS common source amplifier - Calculating R_{OUT}



Figure 1.7: NMOS common source amplifier - Calculating the voltage gain without R_{sig} and R_L



Figure 1.8: NMOS common source amplifier - Calculating the voltage gain with R_{sig} and R_L

1.2 SPICE simulations

1.2.1 DC simulation - Operating Point

```
* NMOS amplifier - DC analysis
******************************************************************************
* 2st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy *
*                       Pietro Prandini - mat. 1097752                        *
*                                                                             *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0 *
* International License. To view a copy of this license, visit                *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                          *
******************************************************************************

* Parameters
.param Vt = 1V
.param Kn = 4m
.param lambda = 0

* NMOS model
.model NMOS NMOS VT0 = Vt KP = Kn LAMBDA = lambda

* Resistances
RG1 VDD G 2.19550MEG
RG2 G 0 1097752
RD VDD D 8K
RS S 0 7K

* Transistors
M1 D G S S NMOS

* Initial conditions
.ic V(VDD) = 15V

* Analysis
.op

.END
```

The results confirm the DC analysis.

```
--- Operating Point ---

V(vdd):    15          voltage
V(g):      5.00001         voltage
V(d):      11          voltage
V(s):      3.50001         voltage
Id(M1):    0.000500001     device_current
Ig(M1):    0           device_current
Ib(M1):    -7.50999e-012   device_current
Is(M1):    -0.000500001    device_current
I(Rs):     0.000500001     device_current
I(Rd):     0.000500001     device_current
I(Rg2):    4.55477e-006    device_current
I(Rg1):    4.55477e-006    device_current
```

1.2.2 AC simulation - Av, R_{IN} and R_{OUT}

```
* NMOS amplifier - Av, RIN and ROUT
```

```
*******************************************************************************
* 2st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy  *
*                     Pietro Prandini - mat. 1097752                           *
*                                                                              *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0  *
* International License. To view a copy of this license, visit                *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative  *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                          *
*******************************************************************************

* Voltage dependent Current Source
gm*vgs D 0 G 0 2m

* Independent Voltage Source
Vx G 0 DC 0 AC 1097752u sin(0 0.1V 10kHz 0 0 0)

* Resistances
RG1 G 0 2.19550MEG
RG2 G 0 1097752
RD D 0 8k

* Analysis
.tf V(D) Vx

.END
```

The result confirm the analysis.

```
--- Transfer Function ---

Transfer_function:          -16      transfer
vx#Input_impedance:         731834   impedance
output_impedance_at_V(d):            8000     impedance
```

1.2.3 AC simulation - Gv

```
* NMOS amplifier - Gv
*******************************************************************************
* 2st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy  *
*                     Pietro Prandini - mat. 1097752                           *
*                                                                              *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0  *
* International License. To view a copy of this license, visit                *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative  *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                          *
*******************************************************************************

* Voltage dependent Current Source
gm*vgs D 0 G 0 2m

* Independent Voltage Source
Vsig SIG 0 DC 0 AC 1097752u sin(0 0.1V 10kHz 0 0 0)

* Resistances
Rsig SIG G 200k
RG1 G 0 2.19550MEG
RG2 G 0 1097752
RD D 0 8k
RL D 0 8k
```

```
* Analysis
.tf V(D) Vsig

.END
```

The result confirm the analysis.

```
--- Transfer Function ---

Transfer_function:          -6.28296         transfer
vsig#Input_impedance:       931834   impedance
output_impedance_at_V(d):            4000     impedance
```

Chapter 2

NMOS common source amplifier without bypass capacitance



Figure 2.1: NMOS common source amplifier

Designing the common source amplifier of the figure 2.1 .
The MOSFET should have a $V_t = 1V$, a $K_n = 4mA/V$ and a $\lambda = 0$.
Other requested parameters are: $I_{DQ} = 0.5mA$, $V_S = 3.5V$, $V_D = 11V$, $V_{DD} = 15V$ and $R_{G2} = 1097752\Omega$.

2.1 Analytic solution

The figure 2.2 represents the circuit for the DC analysis.
It is the same circuit analysed on the section 1.1.1 and so the results of that section are considered also for this section.

Hybrid π model

In this case it is used the hybrid π model (figure 2.3).

R_{IN} from G

Removing the signal, the load and applying a test voltage source as in figure 2.4 it is possible to calculate the input's resistance R_{IN}.
The result is obviously equal to the section 1.1.2's result.



Figure 2.2: NMOS common source amplifier - DC analysis

R_{OUT} from D

Removing the signal, the load and applying a test voltage source as in figure 2.5 it is possible to calculate the output's resistance R_{OUT}.
The result is equal to the section 1.1.2's result.

Voltage Gain - without R_{sig} and R_L

Calculating the gain of the amplifier represented in the figure 2.6.

$$\begin{aligned} v_{in} = v_g &= v_{gs} + v_s && (2.1)\\ &= v_{gs} + v_{gs} g_m R_S && (2.2)\\ &= v_{gs}(1 + g_m R_S) && (2.3) \end{aligned}$$

$$v_o = -g_m v_{gs} R_D \tag{2.4}$$

$$A_v = \frac{v_o}{v_{in}} = \frac{-g_m v_{gs} R_D}{v_{gs}(1 + g_m R_S)} = \frac{-g_m R_D}{(1 + g_m R_S)} = \frac{-2mA/V \cdot 8k\Omega}{1 + 2mA/V \cdot 7k\Omega} = -1.06667 \quad V/V \simeq -1.1 \quad V/V \tag{2.5}$$

Voltage Gain - with R_{sig} and R_L

Calculating the gain of the amplifier represented in the figure 2.7.

$$I_{sig} = \frac{v_{sig}}{R_{sig} + (R_{G1} \parallel R_{G2})} \tag{2.6}$$

$$\begin{aligned} v_{in} = v_g &= v_{sig} - R_{sig} I_{sig} && (2.7)\\ &= v_{sig} - R_{sig} \frac{v_{sig}}{R_{sig} + (R_{G1} \parallel R_{G2})} && (2.8)\\ &= v_{sig} \left(1 - R_{sig} \frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right) && (2.9) \end{aligned}$$



Figure 2.3: NMOS common source amplifier - Hybrid π model

$$v_{gs} = v_g - v_s \tag{2.10}$$

$$v_{gs} = v_{sig}\left(1 - R_{sig}\frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right) - g_m v_g s R_S \tag{2.11}$$

$$v_{gs} + g_m v_{gs} R_S = v_{sig}\left(1 - R_{sig}\frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right) \tag{2.12}$$

$$v_{gs}(1 + g_m R_S) = v_{sig}\left(1 - R_{sig}\frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right) \tag{2.13}$$

$$V_{gs} = v_{sig}\left(1 - R_{sig}\frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right)\frac{1}{1 + g_m R_S} \tag{2.14}$$

$$v_o = -g_m v_{gs}(R_D \parallel R_L) \tag{2.15}$$

$$= -g_m v_{sig}\left(1 - R_{sig}\frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right)\frac{1}{1 + g_m R_S}(R_D \parallel R_L) \tag{2.16}$$

$$G_v = \frac{v_o}{v_{sig}} = \frac{-g_m v_{sig}\left(1 - R_{sig}\frac{1}{R_{sig}+(R_{G1} \parallel R_{G2})}\right)\frac{1}{1+g_m R_S}(R_D \parallel R_L)}{v_{sig}} \tag{2.17}$$

$$= -g_m\left(1 - R_{sig}\frac{1}{R_{sig} + (R_{G1} \parallel R_{G2})}\right)\frac{1}{1 + g_m R_S}(R_D \parallel R_L) \tag{2.18}$$

$$= \frac{-g_m}{1 + g_m R_S}\left(1 - \frac{R_{sig}}{R_{sig} + \left(\frac{R_{G1}R_{G2}}{R_{G1}+R_{G2}}\right)}\right)\left(\frac{R_D R_L}{R_D + R_L}\right) \tag{2.19}$$

$$= \frac{-2mA/V}{1 + 2mA/V \cdot 7k\Omega}\left(1 - \frac{200k\Omega}{200k\Omega + \left(\frac{2.19550M\Omega \cdot 1097752\Omega}{2.19550M\Omega + 1097752\Omega}\right)}\right)\left(\frac{8k\Omega \cdot 8k\Omega}{8k\Omega + 8k\Omega}\right) \tag{2.20}$$

$$= -0.41886 \quad V/V \simeq -0.42 \quad V/V \tag{2.21}$$



Figure 2.4: NMOS common source amplifier - Calculating R_{IN}



Figure 2.5: NMOS common source amplifier - Calculating R_{OUT}



Figure 2.6: NMOS common source amplifier - Calculating the voltage gain without R_{sig} and R_L



Figure 2.7: NMOS common source amplifier - Calculating the voltage gain with R_{sig} and R_L

2.2 SPICE simulations

2.2.1 AC simulation without bypass capacitances - Av, R_{IN} and R_{OUT}

```
* NMOS amplifier without bypass capacitances - Av, RIN and ROUT
*****************************************************************************
* 2st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy *
*                       Pietro Prandini - mat. 1097752                        *
*                                                                             *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0 *
* International License. To view a copy of this license, visit                *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                         *
*****************************************************************************

* Voltage dependent Current Source
gm*vgs D S G S 2m

* Independent Voltage Source
Vx G 0 DC 0 AC 1097752u sin(0 0.1V 10kHz 0 0 0)

* Resistances
RG1 G 0 2.19550MEG
RG2 G 0 1097752
RD D 0 8k
RS S 0 7k

* Analysis
.tf V(D) Vx

.END
```

The result confirm the analysis.

```
--- Transfer Function ---

Transfer_function:          -1.06667        transfer
vx#Input_impedance:         731834   impedance
output_impedance_at_V(d):          8000     impedance
```

2.2.2 AC simulation without bypass capacitances - Gv

```
* NMOS amplifier without bypass capacitances - Gv
*****************************************************************************
* 2st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy *
*                       Pietro Prandini - mat. 1097752                        *
*                                                                             *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0 *
* International License. To view a copy of this license, visit                *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                         *
*****************************************************************************

* Voltage dependent Current Source
gm*vgs D S G S 2m

* Independent Voltage Source
Vsig SIG 0 DC 0 AC 1097752u sin(0 0.1V 10kHz 0 0 0)

* Resistances
```

```
Rsig SIG G 200k
RG1 G 0 2.19550MEG
RG2 G 0 1097752
RD D 0 8k
RL D 0 8k
RS S 0 7k

* Analysis
.tf V(D) Vsig

.END
```

The result confirm the analysis.

```
--- Transfer Function ---

Transfer_function:          -0.418864          transfer
vsig#Input_impedance:       931834    impedance
output_impedance_at_V(d):             4000       impedance
```